

V SYNTAXE BNF* POUR LE LCRC

Nous décrivons dans cette section la syntaxe du Langage Commun de Représentation des Connaissances (LCRC) que nous avons mis au point. Il sert à interfacier les différents modules d'une plate-forme d'aide à la description, à la classification et à la détermination des objets biologiques. Les expressions LCRC se trouvent dans des fichiers ASCII (un fichier pour les attributs, un pour les objets, un pour les valeurs d'attributs et un pour les cas).

Le langage LCRC est à base de rubriques commençant par des mots clés (ce qui permet aux différents outils de ne pas interpréter les rubriques qui ne leur sont pas destinées) et est facilement extensible par l'ajout de nouvelles rubriques. Il permet d'associer des informations aux objets, aux attributs et aux valeurs d'attributs.

V-1 Notes sur les grammaires BNF

Les parenthèses apparaissent telles quelles dans le texte. [, {, <, *, + sont des marques syntaxiques de la grammaire BNF. <> indique une rubrique détaillée plus loin. [] indique que le contenu est optionnel, {} sont des délimiteurs pour borner leur contenu. Ils peuvent être suivies de * ou de + : * indique que ce qui est entre {} apparaît 0 ou plusieurs fois, + au moins une fois et peut-être plusieurs fois. A l'intérieur de {} et [], une barre verticale indique des choix mutuellement exclusifs.

En résumé : {x}* signifie 0 ou plusieurs occurrences de x,
{x}+ une ou plusieurs occurrences de x,
[x] 0 ou 1 occurrence de x,
{x | y} soit x soit y.

* Forme de Backus-Naur.

V-2 Les définitions LCRC du modèle descriptif

V-2.1 Syntaxe des objets

```

<objet> ::= (defobject <nom d'objet>
              [(superObject <nom d'objet>+)]
              [(subObject <nom d'objet>+)]
              [(subparts <nom d'objet>+)]
              [(part-of <nom d'objet>+)]
              [(relations <nom d'attribut>+)]
              [(slots <nom d'attribut>]
              [(question <string>)]
              [(string <string>)])

```

```

ex: (defobject tache-sur-feuilles (sup tache)
      (slots nombre confluence variabilite couleur)
      (string "tache sur feuilles"))

```

V-2.2 Syntaxe des attributs

```

<attribut> ::=
  (defslot <nom d'attribut> <nom d'objet>
    { nominal | ordinal | integer | real | relation }
    (range <range>)
    (cardinal <cardinal>)
    [(question <string>)]
    [(string <string>)]
    [<additional statement>])

```

```

<range> ::= { <nominal range> | <numerical range> | <relation range> }

```

```

<nominal range> ::= <value>+

```

```

<numerical range> ::= <value> <value>

```

- :: Ce sont les valeurs possibles de l'attribut. Pour les numériques, la première
- :: valeur est la borne inférieure et la seconde la borne supérieure. * veut dire ici
- :: l'infini (cas particulier par rapport à la marque syntaxique de la grammaire).
- :: Ex: (defslot taille integer (range 0 *) (cardinal 1))

<relation range> ::= <nom d'objet>+

<cardinal> ::= { 1 | <card inf> <card sup> | <card inf> * }

:: * signifie n'importe quel nombre de valeurs au dessus de <card inf>.

<additional statement> ::= <keyword> <definition>

<keyword> ::= <symbol>

<definition> ::= <any ASCII character>

V-2.3 Syntaxe des valeurs

<valeur> ::= (**defvalue** <nom de valeur>

[(**SuperValue** <nom de valeur>)]

[(**SubValue** <nom de valeur>+)]

[(**String** <string>)])

Note : Une valeur peut être soit la valeur d'un attribut d'un objet, soit un diagnostic.

V-3 Les descriptions LCRC du questionnaire

V-3.1 Syntaxe des cas

<cas> ::= (**defcase** [<numéro de cas>] [<nom du cas>] [<commentaire>]

[<diagnostic>] ; un cas sans diagnostic peut être utilisé en auto-consultation

<description>)

<numéro de cas> ::= (**number** <positive integer>)

<nom de cas> ::= (**name** <string>)

<diagnostic> ::= (**diagnosis** <nom de diagnostic>+)

<commentaire> ::= (**comment** <string>)

<description> ::= (**description**

(**case** <description d'objet>

{(<nom d'objet> [: <identificateur>] <description d'objet>))*

:: l'identificateur (nom d'objet) est nécessaire lorsqu'il y a plusieurs objets de même type.

```
<description objet> ::=
    [(subparts {<object> | (<object> <identificateur>+)} |
    (<object> ?) | (<object> none)+)]
```

```
<attribut>*
```

:: **none** dénote la liste vide (c'est-à-dire un marqueur indiquant qu'il n'y a pas d'objet de ce type)

```
<attribut> ::= (<nom d'attribut> {<value> | ?})
:: ? dénote l'inconnu
```

```
<valeur d'attribut> ::= {<valeur simple> | (<valeur simple>+)} | none
<valeur simple> ::= {symbole | <valeur numérique>}
```

```
<string> ::= "<lettre alpha-numérique sans guillemets>"
```